

5 December 1986

Dear C-Interface Owner,

This is an update to the software that originally came with your C-Interface. It contains two improvements: one is a patch to the commercially available PaintMASTER program that allows you to print pictures produced with this game-controller-driven color graphics design software. The other is a program that patches Smartbasic so that you can print directly from Basic using normal PRINT statements, and list out programs without first SAVEing them.

The first change will allow you to print directly from Basic without using the special GOSUB routines previously required and without building up strings out of the characters to be printed. You can simply use the normal PRINT commands in your program, or in any SmartBasic program you can buy that isn't affected by the different LOMEM value. This change will also allow you to send the output of commands like LIST and CATALOG to the printer, so you will not have to save your program and run comprint to print it out: you can just type PR#2: LIST: PR#0 at any time.

To install this update, type in the program on the last page, and save it on the C-Interface tape under a name like basicprint. The subroutine at its end can be copied from the context program already on the tape, or it can be left out if you do not have a Commodore compatible printer. To use this function after you have loaded your SmartBasic tape, type LOMEM:40000: run basicprint, much like you run graphprint, and answer the question the program asks you about your printer. From that time on, you can send characters to your printer just like you would send it to the Adam printer; except instead of PR#1, you use PR#2 for a Commodore compatible printer, or PR#3 for a printer connected to the Centronics Port. You also can still give the PR#1 command to use the Adam printer.

The basicprint program uses the end of the original CDRIVER print buffer. Because of this, the original software can occasionally destroy it and cause Adam's keyboard to lock up the next time you try to use PR#2. To avoid this, run basicprint again after using the original software if you plan to use PR#2 or PR#3 before turning off your computer.

PaintMASTER uses your game controller as a paintbrush, a pointer to where simple shapes can be automatically drawn, a pointer to where parts of a picture can be moved or duplicated, and a program controller that selects symbols of the operation you want to perform (icons) like the mouse-driven software on computers such as the Apple Macintosh. It will also save and load pictures to tape or disk. Available for around \$25 from several companies that carry Adam software, it can also be bought from the originator, Strategic Software, 424 East John Street, Lindenhurst, NY 11752.

To use it with the C-Interface, you have to make changes on the PaintMASTER tape. First, move CDRIVER onto this tape by loading it into your Adam from the C-Interface tape (run graphprint will do this), and then typing bsave CDRIVER,A27986,L1481 with the PaintMASTER tape in Adam. Then you must change the main PaintMASTER program so it will call CDRIVER. To do this, first rename the original file to save it: rename PM.OBJ,oldpm (PM.OBJ must be in capital letters). Then load it in, making sure LOMEM is small enough: LOMEM:30000: load oldpm. Then type in the following statements:

```
15 GOSUB 21600
21000 t=13: u=20: v=61: GOSUB 200: a$="CONT ERASE PRINT": GOSUB 500
21010 IF p6>118 THEN GOSUB 21600: CALL 28647: TEXT: NEW: END
21600 PRINT CHR$(4);"BLOAD CDRIVER": RETURN
```

The call 28647 statement is for Black & White printing; for color, change this to call 28690. Finally, save PM.OBJ will store these changes.

To print a picture, select the erase function of PaintMASTER while the picture you want to print is on the screen. Then move the pointer to the word PRINT and press the game controller trigger. The picture will then print out. Caution: like the erase function, printing will destroy the picture. If you want to save it, do this before you print it out.

You can also eliminate the side and bottom borders from the picture, which looks better and saves time, ribbon, and wear on the printer. To do this, just before you save CDRIVER, change it with the following POKes that are documented in your C-Interface manual:

```
POKE 28673,50: rem Bottom border for Black & White
POKE 28740,50: rem Bottom border for Color
POKE 28857,4: rem Left border
POKE 28874,31: rem Right border
```

We hope these improvements make your C-Interface easier to use, and allow you to do even more with it than you have done in the past.

Sincerely,

Thomas J. Golab
Capital Software

```

10 REM Program to set up C-Interface to print directly from Basic
11 LOMEM :29723
12 PRINT CHR$(4); "BLOAD CDRIVER"
13 CALL 28325: CALL 29460
20 DATA 22,0,95,33,27,115,25,126
25 DATA 6,17,24,2,6,0,42,134
30 DATA 111,119,35,34,134,111,254,13
35 DATA 40,9,17,223,108,183,237,82
40 DATA 125,180,192,221,229,253,229,197
45 DATA 237,115,122,111,42,122,111,43
50 DATA 43,34,122,111,205,203,114,205
55 DATA 20,115,241,254,17,32,3,205
60 DATA 235,108,253,225,221,225,201,237
65 DATA 67,66,109,205,195,109,195,143,109
70 REM Add new code in C-Interface print buffer
80 sum = 0: FOR adrs = 27871 TO 27951
90 READ byte: POKE adrs, byte: sum = sum+byte
100 NEXT adrs
110 IF sum = 9740 THEN PRINT "New code checksum is correct.": GOTO 150
120 PRINT "New code checksum is wrong-"
130 PRINT "Correct mistakes in data stmts"
140 END
150 REM Patch original C-Interface code
160 POKE 28044, 195: POKE 28045, 38: POKE 28046, 109
200 REM Patch Basic to change PR#2 and PR#3 functions
210 POKE 16217, 223: POKE 16218, 108
220 POKE 16219, 235: POKE 16220, 108
250 PRINT: PRINT "Enter 2 for PR#2 function-"
255 PRINT "3 to use only Centronics port"
257 INPUT "PR#3 function "; a$
260 IF a$ <> "2" THEN 290
265 PRINT: PRINT "PR#2 will send characters to"
266 PRINT "Commodore type (most) printers"
270 GOSUB 45600
280 PR #2: PRINT CHR$(17); : PR #0: GOTO 300
290 LOMEM :29467
300 PRINT: PRINT "PR#3 will send characters to"
310 PRINT "a printer connected to the"
320 PRINT "optional Centronics Port."
330 END
45599 REM Subroutine to initialize ascii to petascii conversion
45600 FOR i = 0 TO 255
45610 POKE 29467+i, i: NEXT i
45620 FOR i = 65 TO 90
45630 POKE 29467+i, i+32: NEXT i
45640 FOR i = 97 TO 127
45650 POKE 29467+i, i-32: NEXT i
45652 POKE 29467+92, 191: POKE 29467+95, 228
45654 POKE 29467+96, 188: POKE 29467+123, 179
45656 POKE 29467+124, 125: POKE 29467+125, 235
45658 POKE 29467+126, 177: POKE 29467+127, 174
45660 REM printer control codes
45662 POKE 29467+19, 18: POKE 29467+20, 146
45664 POKE 29467+28, 18: POKE 29467+28, 18: POKE 29467+29, 146
45666 POKE 29467+30, 18: POKE 29467+31, 146
45690 RETURN

```